UNITED STATES ENVIRONMENTAL PROTECTION AGENCY
WASHINGTON, D.C. 20460

OFFICE OF CHEMICAL SAFETY AND POLLUTION PREVENTION

MEMORANDUM

Date: September 12, 2018

SUBJECT: ***cis*-Permethrin**: Statistical Analysis of PBPK Simulated Data for DDEF

PC Code: 109701
Decision No.: 543196
Petition No.: N/A
Risk Assessment Type: N/A
TXR No.: N/A
MRID No.: N/A

DP Barcode: 448761
Registration No.: N/A
Regulatory Action: Registration Review
Case No.: N/A
CAS No.: 52645-53-1
40 CFR: N/A

FROM: Christine Cai, Statistician
Chemistry and Exposure Branch
Health Effects Division (7509P)

James Nguyen, Mathematical Statistician
Chemistry and Exposure Branch
Health Effects Division (7509P)

THROUGH: David Miller, Branch Chief
Chemistry and Exposure Branch
Health Effects Division (7509P)

TO: Cecilia Tan, Ph.D., Senior Physical Scientist
IO, Health Effects Division (7509P)

and

Evisabel Craig, Ph.D., Toxicologist
Risk Assessment Branch VI
Health Effects Division (7509P)

I. EXECUTIVE SUMMARY

The peak concentration (Cmax) of cis-permethrin in blood and in brain data were reasonably assumed to follow lognormal distributions and statistical analyses did not suggest any gross violations of this assumption. Therefore, mixed-effects models were used to analyze the log-transformed data. The ratios of peak concentration (Cmax) of cis-permethrin in blood or in brain between young age (i.e., 6-month, 1-year, 5-year, and 19-year) groups and the adult group (25-

year) were estimated using the results from the mixed-effect model analyses. Dunnett's test was used to adjust the results from multiple comparisons.

With respect to the **blood plasma** analysis: Except for the 19-year group which had a Cmax in blood plasma that was not significantly different from the adult group (estimated difference = 3% (adjusted 95% CI = (-2.6%, 8.8%))), the Cmax for blood plasma in all 3 of the youngest age groups (6-month, 1-year, and 5-year) were significantly different (higher) than the adult group Estimated differences were +18.1% (adjusted 95% CI = (12.3%, 24.3%)) for the 6 month group; +14.7% (adjusted 95% CI = (9.0%, 20.8%)) for the 1 year group, and +7.0% (adjusted 95% CI=(1.6%, 12.7%) for the 5 year group.

With respect to the **brain** analysis: Except the 5-year group, which had Cmax in the brain significantly different than that of the adult group (estimated difference = -6.3% (adjusted 95% CI = (-12%, -0.2%))), the Cmax of 6-month, 1-year, and 19-year groups were not significantly different from the adult group. The estimated differences were -4.6% (adjusted 95% CI = (-10.4%, 1.6%) for the 6-month group; -5.2% (adjusted 95% CI = (-11%, 1%) for the 1 year group; and +3.1% (adjusted 95% CI = (-3.1%, 9.8%) for the 19 year group.

However, given the large sample size associated with this data (1000 simulations), a small and unimportant difference might be found to be statistically significant. Therefore, biological considerations should be weighed in an assessment in addition to statistical significance as part of the process of determining whether the Cmax of younger age groups were meaningfully or substantively different from the adult group.

II. BACKGROUND

To replace default uncertainty factors, which are 3.16X for PK (pharmacokinetic), 3.16X for PD (pharmacodynamic), and 10X as combined, PBPK modelers and toxicologists are interested in calculating intra-species DDEF (data-derived extrapolation factor). DDEF has a PK component and a PD component. The DDEF for PK is the ratio of: (the median of the most sensitive age group)/(the median of the adult group). A PBPK (in other cases, PBPK/PD) model is used to estimate the DDEF.

From a PBPK model of *cis*-permethrin (CPM), ScitoVation provided Monte Carlo simulation results in five excel spreadsheets for OPP HED (Health Effects Division) review. Each spreadsheet contains 1000 predicted Cmax (peak concentration) values in brain and 1000 predicted Cmax values in blood plasma, from one of five specific age groups: 6-month, 1-year, 5-year, 19-year, and 25-year (considered adults). There is no *a priori* information about the distributions of Cmax data, but they are most likely normal or log-normal because the distributions of parameters are either normal or log-normal.

An HED PBPK modeling expert (Dr. Cecelia Tan) and toxicologists requested statistical support to determine DDEFs for PK in blood plasma and brain from simulated CPM Cmax data, including 1) estimates of the following ratios and their confidence bounds:

- median of 6-month/median of 25-year old;
- median of 1-year old /median of 25-year old;

- median of 5-year /median of 25-year old, and
- median of 19-year / median of 25-year old;

and 2) whether the associated DDEFs for PK (i.e., the above ratios) are statistically different from 1.0 (representing equality) in blood plasma and brain compartments.

Statistical significance and biological significance are two different concepts. Statistical significance is sample-size dependent and large samples sizes will almost invariably produce statistically significant differences that many may consider of inconsequential practical or substantive effect. Given the large sample size such as that found in this simulated data and the likely tendency for minor and substantively inconsequential differences to appear as "statistically significant", CEB recommends that biological significance receive substantial weighting with respect to any conclusions about whether Cmax estimates are different between the various young age groups and the adult group.

III. STATISTICAL METHODS

Scatter-plot and boxplots were created to visually assess the distributions of peak concentration (Cmax) of CPM in blood and brain of the groups. From the visual assessment and the **Kolomogorov-Smirnov** test, the Cmax data were reasonably assumed to be lognormally distributed, and log-transformation was applied to both blood and brain Cmax data. Bartlett's test was used to determine whether the variances were equal among age groups.

Mixed-effects models were used to analyze the log-transformed Cmax data. Since the Bartlett's test indicated the variances of Cmax in blood plasma were significantly different among age groups, a mixed model with heterogeneous variance was used to analyze Cmax in blood plasma. The ratios of Cmax of CPM in blood or in brain between young age (6-month, 1-year, 5-year, and 19-year) groups and adult group (25-year) were estimated using the results from the mixed-effect models. Regression diagnostics were performed to evaluate the model fit: specifically, the assumption of normality in the data. Dunnett's test was used to adjust the results from multiple comparisons.

All the data analyses were performed using SAS 9.4. The SAS code is available in the Appendix: SAS Code.

IV. SUMMARY RESULTS

Tables 1 and 3 show the estimated median Cmax values and their associated confidence intervals for plasma and brain, respectively, while Tables 2 and 4 provide the estimated median Cmax ratios comparing the four age (younger) groups to adults.

Table 1: Estimated Median (95% confidence interval) of Cmax in blood plasma

Age	Estimated Median (95% CI)
BloodP6M	31433.3 (30663.5, 32222.4)
BloodP1Y	30530.5 (29744.3, 31337.6)
BloodP5Y	28469.0 (27721.0, 29237.2)
BloodP19Y	27404.4 (26573.4, 28261.3)
BloodP25Y	26609.7 (25755.1, 27492.7)

Table 2: Estimated median ratios (95% CI) in blood plasma

Group	Referenced Group	Est. Median Ratio in Brain	95% CI	P-value	Dunnett adj. 95% CI	Dunnett adj. p-value
BloodP6M	BloodP25Y	1.181	(1.134, 1.231)	<.001	(1.123, 1.243)	<.001
BloodP1Y	BloodP25Y	1.147	(1.100, 1.196)	<.001	(1.090, 1.208)	<.001
BloodP5Y	BloodP25Y	1.070	(1.026, 1.116)	0.002	(1.016, 1.127)	0.006
BloodP19Y	BloodP25Y	1.030	(0.985, 1.077)	0.199	(0.974, 1.088)	0.475

Table 3: Estimated Median (95% Confidence Interval) of Cmax in brain

Age	Estimated Median (95% CI)
Brain6M	1419.0 (1369.3, 1470.6)
Brain1Y	1410.5 (1361.1, 1461.7)
Brain5Y	1393.8 (1345.0, 1444.4)
Brain19Y	1534.5 (1480.7, 1590.2)
Brain25Y	1487.7 (1435.6, 1541.8)

Table 4: Estimated median ratio (95% Confidence Interval) in brain						
Group	**Referenced Group**	**Est. Median Ratio in Brain**	**95% CI**	**P-value**	**Dunnett adj. 95% CI**	**Dunnett adj. p-value**
Brain6M	Brain25Y	0.954	(0.907, 1.003)	0.066	(0.896, 1.016)	0.200
Brain1Y	Brain25Y	0.948	(0.901, 0.997)	0.038	(0.890, 1.010)	0.122
Brain5Y	Brain25Y	0.937	(0.891, 0.985)	0.011	(0.880, 0.998)	0.039
Brain19Y	Brain25Y	1.031	(0.981, 1.085)	0.230	(0.969, 1.098)	0.566

Based on the above summary tables and analyses, the percent differences of DDEFs for PK in brain from simulated CPM Cmax data between the young groups vs. reference adult group range from a statistically significant -6.3% (adjusted p=0.039) in the 5-year group to a non-significant +3.1% in the 19-year old group. For blood plasma, the differences when comparing the young groups to the adult group range from a statistically significant +18.1% in the 6-month group to a non-statistically significant +3% in the 19-year old group.

V. OTHER RESULTS

1. Original data were not normally distributed (Figure 1-4).

Figure 1: Scatter plot of blood plasma data



Figure 2: Boxplot of blood plasma data



Figure 3: Scatter plot of brain data



Figure 4: Boxplot of brain data



2. Boxplots (Figures 5 and 6) and distribution test results (Kolmogorov-Smirnov, Cramer-von Mises, and Anderson-Darling) confirmed log transformation is appropriate in this case. Tables 5 and 6 are goodness-of-fit results from Kolmogorov-Smirnov tests.

Figure 5: Box plot of the log transformed blood plasma data.



Figure 6: Box plot of the log transformed brain data.



Table 5: Distributions comparison (Kolomogorov-Smirnov) of original and log-transformed blood plasma data

Goodness-of-Fit Tests for Normal Distribution				
Kolmogorov-Smirnov	**Statistic**		**p Value**	
1YBP (Original data)	**D**	0.04623381	**Pr > D**	<0.010
1YBP (Log transformed)	**D**	0.01599596	**Pr > D**	>0.500
5YBP (Original data)	**D**	0.05270999	**Pr > D**	<0.010
5YBP (Log transformed)	**D**	0.02047940	**Pr > D**	>0.250
6MBP (Original data)	**D**	0.04711920	**Pr > D**	<0.010
6MBP (Log transformed)	**D**	0.01915021	**Pr > D**	>0.250
19YBP (Original data)	**D**	0.0687389	**Pr > D**	<0.010
19YBP (Log transformed)	**D**	0.02752808	**Pr > D**	0.031
25YBP (Original data)	**D**	0.1064224	**Pr > D**	<0.010
25YBP (Log transformed)	**D**	0.01513164	**Pr > D**	>0.500

Table 6: Distributions comparison (Kolomogorov-Smirnov) of original and log-transformed brain data

Goodness-of-Fit Tests for Normal Distribution				
Kolmogorov-Smirnov	**Statistic**		**p Value**	
1YBP (Original data)	**D**	0.1091619	**Pr > D**	<0.010
1YBP (Log transformed)	**D**	0.02270370	**Pr > D**	0.164
5YBP (Original data)	**D**	0.0987578	**Pr > D**	<0.010
5YBP (Log transformed)	**D**	0.02474018	**Pr > D**	0.089
6MBP (Original data)	**D**	0.1268214	**Pr > D**	<0.010
6MBP (Log transformed)	**D**	0.02360778	**Pr > D**	0.127
19YBP (Original data)	**D**	0.1184245	**Pr > D**	<0.010
19YBP (Log transformed)	**D**	0.01919408	**Pr > D**	>0.250
25YBP (Original data)	**D**	0.1270446	**Pr > D**	<0.010
25YBP (Log transformed)	**D**	0.02267195	**Pr > D**	0.165

3. Tests for homogeneity of variance (Bartlett's test) indicated the variance of log-transformed Cmax in blood plasma were significantly different among age groups ($p < 0.0001$).

4. To accommodate heterogeneous variance of Cmax in blood plasma among the age groups, a mixed effects model with option of unequal variance was used to analyze the log-transformed Cmax data in blood plasma.

5. The regression diagnostics showed that the residuals were close to normally distributed, which satisfied the assumption of the models.

Figure 7: Regression Diagnostics of mixed model fitting blood plasma Cmax data.



Figure 8: Regression diagnostics of the mixed model fitting brain Cmax data



APPENDIX: SAS CODE

```
/*************************************************************************
* Analysis of OPP PBPK simulation data for cis-permethrin                *
* Statistician: Christine Cai                                            *
* Purpose:      Analyze data                                             *
*               Answer questions                                         *
* Date:         07/30/2018                                               *
* Data:         PBPK simulation data provided by Cecilia Tan, OPP        *
*************************************************************************/

options nodate nonumber linesize=100 pagesize=100 formdlim="=";
%let junks=C:\Users\JNguyen\Desktop\Junks;

libname PYR 'F:\Cecilia Tan';
Title1 'Cis-permethrin';

/*Import and merge data*/
option Mprint;
%Macro import (to_import=, out_file=, newname1=, newname2=);
proc import
        datafile=&to_import
        out=&out_file (rename=(Brain=&newname1 Blood_Plasma=&newname2))
        DBMS=csv
        REPLACE;
        datarow=2;
run;
proc print data=&out_file (obs=25);
run;
%Mend import;

%import (to_import='F:\Cecilia Tan\CPM-1Y.csv', out_file=PYR.CPM1Y, newname1=Brain1Y,
newname2=BloodP1Y);
%import (to_import='F:\Cecilia Tan\CPM-6M.csv', out_file=PYR.CPM6M, newname1=Brain6M,
newname2=BloodP6M);
%import (to_import='F:\Cecilia Tan\CPM-5Y.csv', out_file=PYR.CPM5Y, newname1=Brain5Y,
newname2=BloodP5Y);
%import (to_import='F:\Cecilia Tan\CPM-19Y.csv', out_file=PYR.CPM19Y, newname1=Brain19Y,
newname2=BloodP19Y);
%import (to_import='F:\Cecilia Tan\CPM-25Y.csv', out_file=PYR.CPM25Y, newname1=Brain25Y,
newname2=BloodP25Y);

data PYR.CPM6M;
        set PYR.CPM6M;
        rename Dose___14_115_mg_kg_day=Sample;
run;
proc print data=PYR.CPM6M (obs=25);
run;

data PYR.CPM_All;
        merge PYR.CPM6M PYR.CPM1Y PYR.CPM5Y PYR.CPM19Y PYR.CPM25Y ;
        by Sample;
        drop Var1;
run;
proc print data=PYR.CPM_All(obs=25);
run;


/*Data distribution and transformation*/

* 1. Exploring data distribution with histogram and ggplot;
%Macro Histogram_GGPlot(Age=,outfile=);
ods rtf file =&outfile;
ods graphics on;
proc univariate data=PYR.CPM_All noprint;
        var &Age;
        histogram/ normal
                lognormal(theta=est zeta=est sigma=est)
                exponential(theta=est sigma=est)
                gamma (theta=est alpha=est sigma=est)
```

```
                weibull(c=est sigma=est)
                odstitle = "&Age";
        inset n mean(5.3) std='Std Dev'(5.3) skewness(5.3)
           / pos = ne  header = 'Summary Statistics';
        qqplot/normal odstitle ="&Age";
        qqplot/lognormal (sigma=est theta=est zeta=est) odstitle ="&Age";
        qqplot/exponential(theta=est sigma=est) odstitle ="&Age";
        qqplot/gamma (theta=est alpha=est sigma=est) odstitle ="&Age";
        qqplot/weibull(c=est sigma=est) odstitle ="&Age";
run;
ods graphics off;
ods rtf close;
%Mend Histogram_GGPlot;

%Histogram_GGPlot (Age=Brain6M, outfile="&junks\Brain6M.rtf");
%Histogram_GGPlot (Age=Brain1Y, outfile="&junks\Brain1Y.rtf");
%Histogram_GGPlot (Age=Brain5Y, outfile="&junks\Brain5Y.rtf" );
%Histogram_GGPlot (Age=Brain19Y, outfile="&junks\Brain19Y.rtf");
%Histogram_GGPlot (Age=Brain25Y, outfile="&junks\Brain25Y.rtf");
%Histogram_GGPlot (Age=BloodP6M, outfile="&junks\BloodP6M.rtf");
%Histogram_GGPlot (Age=BloodP1Y, outfile="&junks\BloodP1Y.rtf");
%Histogram_GGPlot (Age=BloodP5Y, outfile="&junks\BloodP5Y.rtf");
%Histogram_GGPlot (Age=BloodP19Y, outfile="&junks\BloodP19Y.rtf");
%Histogram_GGPlot (Age=BloodP25Y, outfile="&junks\BloodP25Y.rtf");

*2. Focus on Normal vs Lognormal with plots;
%Macro Histogram_GGPlot_N_LogN(Age=,outfile=);
ods rtf file =&outfile;
ods graphics on;
proc univariate data=PYR.CPM_All noprint;
        var &Age;
        histogram/ normal
                lognormal(theta=est zeta=est sigma=est)
                odstitle = "&Age";
        inset n mean(5.3) std='Std Dev'(5.3) skewness(5.3)
           / pos = ne  header = 'Summary Statistics';
        qqplot/normal odstitle ="&Age";
        qqplot/lognormal (sigma=est theta=est zeta=est) odstitle ="&Age";
run;
ods graphics off;
ods rtf close;
%Mend Histogram_GGPlot_N_LogN;

%Histogram_GGPlot_N_LogN (Age=Brain6M, outfile="&junks\Brain6M.rtf");
%Histogram_GGPlot_N_LogN (Age=Brain1Y, outfile="&junks\Brain1Y.rtf");
%Histogram_GGPlot_N_LogN (Age=Brain5Y, outfile="&junks\Brain5Y.rtf" );
%Histogram_GGPlot_N_LogN (Age=Brain19Y, outfile="&junks\Brain19Y.rtf");
%Histogram_GGPlot_N_LogN (Age=Brain25Y, outfile="&junks\Brain25Y.rtf");
%Histogram_GGPlot_N_LogN (Age=BloodP6M, outfile="&junks\BloodP6M.rtf");
%Histogram_GGPlot_N_LogN (Age=BloodP1Y, outfile="&junks\BloodP1Y.rtf");
%Histogram_GGPlot_N_LogN (Age=BloodP5Y, outfile="&junks\BloodP5Y.rtf");
%Histogram_GGPlot_N_LogN (Age=BloodP19Y, outfile="&junks\BloodP19Y.rtf");
%Histogram_GGPlot_N_LogN (Age=BloodP25Y, outfile="&junks\BloodP25Y.rtf");

*3. Normality test on log transformed data;
data PYR.CPM_LogT;
        set PYR.CPM_All;
        LogBrain6M=log(Brain6M);
        LogBrain1Y=log(Brain1Y);
        LogBrain5Y=log(Brain5Y);
        LogBrain19Y=log(Brain19Y);
        LogBrain25Y=log(Brain25Y);
        LogBloodP6M=log(BloodP6M);
        LogBloodP1Y=log(BloodP1Y);
        LogBloodP5Y=log(BloodP5Y);
        LogBloodP19Y=log(BloodP19Y);
        LogBloodP25Y=log(BloodP25Y);
        drop Brain6M BloodP6M Brain1Y BloodP1Y Brain5Y BloodP5Y Brain19Y BloodP19Y Brain25Y
BloodP25Y;
run;
proc print data=PYR.CPM_LogT(obs=25);
```

```
run;

ods rtf file = "&junks\Normality Test on logTransformed Data.rtf";
ods graphics on;
proc univariate data=PYR.CPM_LogT normaltest plots;
        var LogBrain6M LogBrain1Y LogBrain5Y LogBrain19Y LogBrain25Y LogBloodP6M LogBloodP1Y
        LogBloodP5Y LogBloodP19Y LogBloodP25Y;
        output out=LogTVar var=VarLB6M var=VarLB1Y var=VarLB5Y var=VarLB19Y var=VarLB25Y
                           var=VarLBP6M var=VarLBP1Y var=VarLBP5Y var=VarLBP19Y var=VarLBP25Y;
        output out=NormalityProbn Probn=ProbnLB6M Probn=ProbnLB1Y Probn=ProbnLB5Y
        Probn=ProbnLB19Y Probn=ProbnLB25Y Probn=ProbnLBP6M Probn=ProbnLBP1Y Probn=ProbnLBP5Y
        Probn=ProbnLBP19Y Probn=ProbnLBP25Y;
run;
proc print data=LogTVar noobs;
        Title2 "Variances of LogTransformated Data";
run;
proc print data=NormalityProbn noobs;
        Title2 "Probability Values for Shapiro-Wilk Normality Test on LogTransformated Data";
run;
ods graphics off;
ods rtf close;

*Blood Plasma;
data PYR.CPM_BloodP_Log;
        set PYR.CPM_LogT;
        keep Sample LogBloodP6M LogBloodP1Y LogBloodP5Y LogBloodP19Y LogBloodP25Y;
run;
proc print data=PYR.CPM_BloodP_Log(obs=25);
run;

proc transpose data=PYR.CPM_BloodP_Log
        out=PYR.CPM_BloodP_LogT (rename=(col1=CMax))
        name=Age;
        by Sample;
        var LogBloodP6M LogBloodP1Y LogBloodP5Y LogBloodP19Y LogBloodP25Y;
run;
proc print data=PYR.CPM_BloodP_LogT(obs=10);
run;

*Brain;
data PYR.CPM_Brain_Log;
        set PYR.CPM_LogT;
        keep Sample LogBrain6M LogBrain1Y LogBrain5Y LogBrain19Y LogBrain25Y;
run;
proc print data=PYR.CPM_Brain_Log(obs=25);
run;

proc transpose data=PYR.CPM_Brain_Log
        out=PYR.CPM_Brain_LogT (rename=(col1=CMax))
        name=Age;
        by Sample;
        var LogBrain6M LogBrain1Y LogBrain5Y LogBrain19Y LogBrain25Y;
run;
proc print data=PYR.CPM_Brain_LogT(obs=10);
run;


**
**BloodP;
ods rtf file = "&junks\BloodPlasmaOutput1.rtf";
title;
title1 Cis-permethrin;
title2;
*Scatter Plots;
data CPM_BloodP;
        set PYR.CPM_All;
        keep Sample BloodP6M  BloodP1Y  BloodP5Y  BloodP19Y BloodP25Y;
run;
proc transpose data=CPM_BloodP
        out=CPM_BloodP_T (rename=(col1=CMax))
        name=Age;
```

```
        by Sample;
        var BloodP6M BloodP1Y BloodP5Y BloodP19Y BloodP25Y;
run;
ods graphics on/ antialiasmax=5000;
PROC sgplot DATA=CPM_BloodP_T;
        scatter y=CMax x=Age /group = Age;
        Xaxis label= "Age Group";
        Yaxis label= "CMax";
        title2 Scatter plot of original blood plasma data;
RUN;
ods graphics off;

*Summary statistics and plots;
Proc means data=CPM_BloodP_T noprint;
        class Age;
        var CMax;
        output out=CPM_BloodP_T_d N=N mean=mean median=median Std=Std;
run;
data CPM_BloodP_T_d;
        set CPM_BloodP_T_d;
        if _type_=0 then Age="Total";
        drop _type_ _freq_;
run;
Proc print data=CPM_BloodP_T_d;
        title2 Summary statistics of original blood plasma data;
run;

PROC sgplot DATA=CPM_BloodP_T_d;
        scatter x=mean y=std/datalabel=age;
        Xaxis label= "Mean";
        Yaxis label= "StDev";
        reg x = mean y =std;
        title2 Scatter plot for mean and StDev of original blood plasma data;
RUN;

*Anova and Dunnett's;
ods graphics on;
proc anova data=CPM_BloodP_T;
        class Age;
        model CMax=Age;
        means Age/hovtest hovtest=bartlett dunnett('BloodP25Y') CLdiff;
        title2 ANOVA for original blood plasma data;
run;
ods graphics off;


*Log transformed data and plots;
ods graphics on/ antialiasmax=5000;
PROC sgplot DATA=PYR.CPM_BloodP_LogT;
        scatter y=CMax x=Age /group = Age;
        Xaxis label= "Age Group";
        Yaxis label= "LogCMax";
        title2 Scatter plot of logtransformed blood plasma data;
RUN;
ods graphics off;

*Summary statistics and plots;
Proc means data=PYR.CPM_BloodP_LogT noprint;
        class Age;
        var CMax;
        output out=CPM_BloodP_LogT_d N=N mean=mean median=median Std=Std;
run;
data CPM_BloodP_LogT_d;
        set CPM_BloodP_LogT_d;
        if _type_=0 then Age="Total";
        drop _type_ _freq_;
run;
proc print data=CPM_BloodP_LogT_d;
        title2 Summary statistics of log transformed blood plasma data;
run;
PROC sgplot DATA=CPM_BloodP_LogT_d;
```

```
        scatter x=mean y=std/datalabel=age;
        Xaxis label= "Mean";
        Yaxis label= "StD";
        reg x = mean y =std;
        title2 Scatter plot of mean and variance for log transformed blood plasma data;
RUN;

*Anova and Dunnett's;
ods graphics on;
proc anova data=PYR.CPM_BloodP_LogT;
        class Age;
        model CMax=Age;
        means Age/hovtest=bartlett dunnett('LogBloodP25Y') CLdiff;
        title2 ANOVA for log transformed blood plasma data;
run;
ods graphics off;
ods rtf close;

ods rtf file = "&junks\BloodP LogT, Prox Mixed, LSmeans, Residual Diagnostic.rtf";
ods graphics on;
ods output Diffs=Diffs lsmeans=lsmeans;
proc mixed data=PYR.CPM_BloodP_LogT plots=all;
        class Age;
        model CMax= Age/residual;
        repeated / group = age;
        lsmeans Age/diff=control("LogBloodP25Y") cl adj=dunnett;
        title2 Blood Plasma;
run;
ods graphics off;
ods rtf close;
*proc print data=Diffs;
*run;


ods rtf file = "&junks\Blood Plasma Age Ratio and Confidence Interval from ProxMixed.rtf"
startpage=no;
Data Lsmeans;
        set Lsmeans;
        lsmeans_CI = trim(left(compress(left(put(exp(Estimate), 10.1)))))||
                        "  "||trim(left(compbl("("||compress(left(put(exp(Lower),10.1))))||
                        ", "||compress(left(put(exp(Upper),10.1))))||")")));
run;
Proc print data = Lsmeans noobs;
        var age lsmeans_CI;
run;
proc datasets nolist;
        delete Lsmeans;
quit;


data PYR.BloodPMixedDiffs;
        set Diffs;
        Age1=substr(Age,4);
        Age2=substr(_Age,4);
        MedianRatio_BloodPlasma=(left(put(exp(Estimate), 6.3)));
        CI_95Pct="("||(left(put(exp(Lower),6.3)))||
                ", "||(left(put(exp(Upper),6.3)))||")";
        P=Probt;
        CI_95Pct_DunnAdj="("||(left(put(exp(AdjLower),6.3)))||
                ", "||(left(put(exp(AdjUpper),6.3)))||")";
        P_DunnAdj=AdjP;
        format P P_DunnAdj pvalue5.;
        Keep Age1 Age2 MedianRatio_BloodPlasma CI_95Pct P CI_95Pct_DunnAdj P_DunnAdj;
run;
proc print data=PYR.BloodPMixedDiffs noobs;
        title2 CMax Ratio in Blood Plasma at Different Ages;
run;
ods rtf close;
```

```
**Brain;
ods rtf file = "C:\Users\CCai\Documents\SASProj\OPP\Pyrethroid\BrainOutput1.rtf";
title;
title1 Cis-permethrin;
title2;
data CPM_Brain;
        set PYR.CPM_All;
        keep Sample Brain6M Brain1Y Brain5Y Brain19Y Brain25Y ;
run;
proc transpose data=CPM_Brain
        out=CPM_Brain_T (rename=(col1=CMax))
        name=Age;
        by Sample;
        var Brain6M Brain1Y Brain5Y Brain19Y Brain25Y;
run;

ods graphics on/ antialiasmax=5000;
PROC sgplot DATA=CPM_Brain_T;
        scatter y=CMax x=Age /group = Age;
        Xaxis label= "Age Group";
        Yaxis label= "CMax";
        title2 Scatter plot of original brain data;
RUN;
ods graphics off;

*Summary statistics and plots;
Proc means data=CPM_Brain_T noprint;
        class Age;
        var CMax;
        output out=CPM_Brain_T_d N=N mean=mean median=median Std=Std;
run;
data CPM_Brain_T_d;
        set CPM_Brain_T_d;
        if _type_=0 then Age="Total";
        drop _type_ _freq_;
run;
PROC sgplot DATA=CPM_Brain_T_d;
        scatter x=mean y=std/datalabel=age;
        Xaxis label= "Mean";
        Yaxis label= "StDev";
        reg x = mean y =std;
        title2 Scatter plot for mean and variance of original brain data;
RUN;

*Anova and Dunnett's;
ods graphics on;
proc anova data=CPM_Brain_T;
        class Age;
        model CMax=Age;
        means Age/hovtest dunnett('Brain25Y') CLdiff;
        title2 ANOVA for original brain data;
run;
ods graphics off;

*Log transformed data and plots;
ods graphics on/ antialiasmax=5000;
PROC sgplot DATA=PYR.CPM_Brain_LogT;
        scatter y=CMax x=Age /group = Age;
        Xaxis label= "Age Group";
        Yaxis label= "CMax";
        title2 Scatter plot of logtransformed brain data;
RUN;
ods graphics off;

*Summary statistics and plots;
Proc means data=PYR.CPM_Brain_LogT noprint;
        class Age;
        var CMax;
        output out=CPM_Brain_LogT_d N=N mean=mean median=median Std=Std;
run;
```

```
data CPM_Brain_LogT_d;
        set CPM_Brain_LogT_d;
        if _type_=0 then Age="Total";
        drop _type_ _freq_;
run;
proc print data=CPM_Brain_LogT_d;
title2 Summary statistics of log transformed brain data;
run;
PROC sgplot DATA=CPM_Brain_LogT_d;
        scatter x=mean y=std/datalabel=age;
        Xaxis label= "Mean";
        Yaxis label= "StDev";
        reg x = mean y =std;
        title2 Scatter plot of mean and variance for log transformed brain data;
RUN;

*Anova and Dunnett's;
ods graphics on;
proc anova data=PYR.CPM_Brain_LogT;
        class Age;
        model CMax=Age;
        means Age/hovtest=bartlett dunnett('LogBrain25Y') CLdiff;
        title2 ANOVA for log transformed brain data;
run;
ods graphics off;

ods rtf close;

ods rtf file = "&junks\Brain LogT, Prox Mixed, LSmeans, Residual Diagnostic.rtf";
ods graphics on;
ods output Diffs=Diffs lsmeans=lsmeans;
proc mixed data=PYR.CPM_Brain_LogT plots=all;
        class Age;
        model CMax= Age/residual;
        lsmeans Age/diff=control("LogBrain25Y") cl adj=dunnett;
        title2 Brain ;
run;
ods graphics off;
ods rtf close;
proc print data=Diffs;
run;

Data Lsmeans;
        set Lsmeans;
        lsmeans_CI = trim(left(compress(left(put(exp(Estimate), 10.1)))))||
                        "  "||trim(left(compbl("("||compress(left(put(exp(Lower),10.1))))||
                        ", "||compress(left(put(exp(Upper),10.1))))||")")));
run;
Proc print data = Lsmeans noobs;
        var age lsmeans_CI;
run;
proc datasets nolist;
        delete Lsmeans;
quit;

ods rtf file = "&junks\Brain Age Ratio and Confidence Interval from ProxMixed.rtf";
data PYR.BrainMixedDiffs;
        set Diffs;
        Age1=substr(Age,4);
        Age2=substr(_Age,4);
        MedianRatio_Brain=(left(put(exp(Estimate), 6.3)));
        CI_95Pct="("||(left(put(exp(Lower),6.3)))||
                ", "||(left(put(exp(Upper),6.3)))||")";
        P=Probt;
        CI_95Pct_DunnAdj="("||(left(put(exp(AdjLower),6.3)))||
                ", "||(left(put(exp(AdjUpper),6.3)))||")";
        P_DunnAdj=AdjP;
        format P P_DunnAdj pvalue5.;
        Keep Age1 Age2 MedianRatio_Brain CI_95Pct P CI_95Pct_DunnAdj P_DunnAdj;
run;
```

```
proc print data=PYR.BrainMixedDiffs noobs;
       title2 CMax Ratio in Brain at Different Ages;
run;
ods rtf close;

ods listing close;
*Output Summary tables for both Blood Plasma and Brain;
ods rtf file = "C:\Users\CCai\Documents\SASProj\OPP\Pyrethroid\Summary.rtf" startpage=no;
title2 CMax Ratio;
proc print data=PYR.BloodPMixedDiffs noobs ;
run;
proc print data=PYR.BrainMixedDiffs noobs;
run;
ods rtf close;
```